



שפת ASP - בדיקת כתובות Email

ניר אדר

מסמך זה הורד מהאתר <http://underwar.livedns.co.il>

אין להפיץ מסמך זה במדיה כלשהי, ללא אישור מפורש מאת המחבר.

מחבר המסמך איננו אחראי לכל נזק, ישיר או עקיף, שיגרם עקב השימוש במידע המופיע במסמך, וכן לנכונות התוכן של הנושאים המופיעים במסמך. עם זאת, המחבר עשה את מירב המאמצים כדי לספק את המידע המדויק והמלא ביותר.

כל הזכויות שמורות לניר אדר

Nir Adar

Email: underwar@hotmail.com

Home Page: <http://underwar.livedns.co.il>

אנא שלחו תיקונים והערות אל המחבר.

בדיקת כתובת מייל הניתנת על ידי המשתמש

כאשר אנו בונים אתר, נרצה לפעמים לקבל מן המשתמש את כתובת ה-Email שלו, על מנת לשלוח לו דואל במועד מאוחד יותר.

מקרים קלאסיים לכך הם רשימת תפוצה של אתר, או משתמשים המבקשים עדכון כאשר מתפרסמת תגובה למשהו שהם פרסמו, וכדו'.

אנחנו, בתור בוני האתר, רוצים לוודא שהכתובת שהכניס המשתמש היא חוקית, היא קיימת, והיא של אותו משתמש שהכניס אותה. (ייתכן שמישהו הכניס כתובת דואל של אדם אחר). אם לא נבצע בדיקות אלו, אז יתכן שנשלח דואל שלא יגיע אל יעדו (עקב טעות הקלדה, למשל, של כתובת המייל) או שיגיע לאדם שלא ביקש כלל לקבל מייל מאיתנו.

הבדיקה שאנו יכולים לבצע יכולה להתבצע במספר דרכים:

1. **הוספת קוד Javascript בצד הלקוח הבדוק את חוקיות ה-Email.** שיטה זו איננה מומלצת, כי אין לנו שליטה האם הדפדפן של המשתמש אכן ביצע את הסקריפט (המשתמש יכול להשתמש בדפדפן ישן, או שאולי הוא כיבה את אפשרות ה-Javascript, או שאולי מדובר במשתמש זדוני שבכוונה נמנע מלבצע את הבדיקה).
2. **בדיקת בצד השרת (למשל בקוד ASP) את חוקיות מבנה הכתובת שהוכנסה** – זו השיטה הנפוצה ביותר, ולרוב היא גם מספיקה. בדומה לבדיקה שנעשית בצד הלקוח – אנו בודקים שמבנה הכתובת שהוכנסה הוא חוקי – כלומר שיש בו רק @ אחד, שקיימים רק סימנים אלפביתיים חוקיים בתוספת נקודות, וכו'. במסמך זה נציג שתי דוגמאות כיצד ניתן לבצע בדיקה זו.
3. **שליחה של אימייל אל הכתובת שהוכנסה, לוידוא נכונות הכתובת** – הרעיון: נשלח דואל אל המשתמש, ונבקש ממנו להכנס לכתובת מסויימת (הייחודית רק לו) לצורך אישור קבלת הדואל. על ידי שימוש בשיטה זו אנו משיגים שתי מטרות: 1. אנו בודקים שמדובר אכן במייל חוקי וקיים (אחרת אף אחד לא יקבל את הדואל, ולא נקבל אישור). 2. אנחנו מקבלים אישור מבעל תיבת ה-Email שהוא אכן מעוניין באימיילים מאיתנו (במידה ולא, הוא לא יחזיר תשובה לאימייל ששלחנו אליו). בדיקה כזו לא תודגם במסמך זה, אולם היא מומלצת ביותר.

בדיקות בצד השרת

עיקר מסמך זה תהיה הצגת דרכים לבדיקת הכתובת שהגיש המשתמש בצד השרת. נציג כעת שתי דרכים לבדוק את הכתובת שהוכנסה בצד השרת. הרעיון בשתייהן זהה, אולם דרך הביצוע שונה.

הדוגמא הראשונה מציגה את בדיקת הכתובת בצורה קלאסית: לולאה העוברת על המחרוזת שהכניס המשתמש בתור כתובת Email, ובדיקה שלה תו תו. הדוגמא השנייה מבצעת את הבדיקה על ידי שימוש בביטוי רגולרי, והחזרת אמת אם יש התאמה בינו ובין הכתובת שהוכנסה.

הקוד שנציג הינו קוד ASP. שתי הפונקציות המוצגות מקבלות פרמטר מסוג מחרוזת, שמייצג כתובת Email. הפונקציות מחזירות 1 אם הכתובת אינה חוקית, ו-0 אם היא חוקית. הקוד מתועד על מנת להסביר מה בעצם אנהנו מנסים לבצע בכל שלב.

דוגמא 1

```

.....
' Function Name:
'   chkEmail
' Description:
'   Checks if given email address is legal
' Parameters:
'   theAddress - Address to check
' Returns:
'   returns 1 for invalid addresses. returns 0 for valid addresses
.....
Function chkEmail(theAddress)
    dim atCnt
    chkEmail = 0

    ' check length
    if len(theAddress) < 5 then
        ' a@b.c should be the shortest an
        ' address could be
        chkEmail = 1

    ' check format
    ' has at least one "@"
    elseif instr(theAddress,"@") = 0 then
        chkEmail = 1

    ' has at least one "."

```

```
elseif instr(theAddress, ".") = 0 then
    chkEmail = 1

' has no more than 3 chars after last "."
elseif len(theAddress) - instrrev(theAddress, ".") > 3 then
    chkEmail = 1

' has no "_" after the "@"
elseif instr(theAddress, "_") <> 0 and _
    instrrev(theAddress, "_") > instrrev(theAddress, "@") then
    chkEmail = 1

else
    ' has only one "@"
    atCnt = 0
    for i = 1 to len(theAddress)
        if mid(theAddress, i, 1) = "@" then
            atCnt = atCnt + 1
        end if
    next

    if atCnt > 1 then
        chkEmail = 1
    end if

    ' check each char for validity
    for i = 1 to len(theAddress)
        if not isnumeric(mid(theAddress, i, 1)) and _
            (lcase(mid(theAddress, i, 1)) < "a" or _
            lcase(mid(theAddress, i, 1)) > "z") and _
            mid(theAddress, i, 1) <> "_" and _
            mid(theAddress, i, 1) <> "." and _
            mid(theAddress, i, 1) <> "@" and _
            mid(theAddress, i, 1) <> "-" then
            chkEmail = 1
        end if
    next

    End If
End Function
```

דוגמא 2

הקוד הבא מבצע משימה זהה לזו שביצע הקוד לעיל. ההבדל הוא שבדוגמא זו אנו משתמשים בביטוי רגולרי כדי לבדוק את כתובת המייל.

ההבדלים בבחירה בין הדוגמאות הם אותם יתרונות בכל מקום בו אנו משתמשים בביטוי רגולרי: ביטוי רגולרי לרוב מובן יותר, יותר מהיר לכתיבה ויותר מהיר לתיקון במקרה של תקלה או שינוי בדרישות – לעומת זאת ביטוי רגולרי פחות יעיל מבדיקה ישירה. הבחירה בין הדוגמאות תלויה במקרה הספציפי ובהעדפותיו של המשתמש.

```

.....
' Function Name:
'   chkEmail
' Description:
'   Checks if given email address is legal
' Parameters:
'   theAddress - Address to check
' Returns:
'   returns 1 for invalid addresses. returns 0 for valid addresses
.....

Function chkEmail(theAddress)
    Dim Reg, MatchResults
    Set Reg = Server.CreateObject("VBScript.RegExp")
    Reg.Global = True
    Reg.IgnoreCase = True
    Reg.Pattern = "\w+([-+.]\w+)*@\w+([-.]\w+)*\.\w+([-.]\w+)*"
    Set MatchResults = Reg.Execute(theAddress)
    If (MatchResults.Count <> 1) Then
        chkEmail = 1
    Else
        If (MatchResults.Item(0).Value <> theAddress) then
            chkEmail = 1
        Else
            chkEmail = 0
        End If
    End If

    Set Reg = Nothing
End Function

```

EOF